JOURNAL OF APPLIED CRYSTALLOGRAPHY

Volume 54 (2021)

Supporting information for article:

Simultaneous measurement of pole figure and residual stress for polycrystalline thin films – ω–φ′ compensated grazing-incidence diffraction in side-inclination mode

Xiaodong Wang

Appendix F

DIFFRAC.TOPAS features a pre-processing step for INP files before kernel treatment, which allows recursive macros in below INP file to simultaneously load and refine hundreds of 1D data as well as analyse and report their refinable parameters. Below INP file can be executed in either DIFFRAC.TOPAS v6 or v7 to generate a 3 column (Chi, Phi, Intensity) pole figure file (.txt format) which can be directly imported in DIFFRAC.TEXTURE v4.1 for further texture analysis.

```
'This INP file performs simultaneous fittings for 1D peak profiles (.xy) of same hkl looped on Phi
then looped on Chi. The data files need be named as "c_n", in which "c" refers to any filename and 'n'
refers to a continuous running number e.g. from 1 to Chi_Num*Phi_Num. Cf: X.Wang(2021) J.Appl.Cryst.

'macro Numeric(n) {#prm a = n; #out a} 'v7 TOPAS.inc already included this macro; uncomment this line
when running this INP in v6.

'Total 324 measured peaks (of 9 Chi tilts * 36 Phi directions)
#prm Chi_Num 9 ' please provide the number of Chi tilts
#prm Phi_Num 36 ' please provide the number of Phi directions
#prm Total_Num =Chi_Num*Phi_Num;

macro load_sequential_data(c,n,n2) '(c=any_filename, n=start_number, n2=finish_number)
{#m_if n>n2; #m_else
xdd c##_##n##.xy local Chi = 10 Round((n-(#out Phi_Num +1)/2)/#out Phi_Num); local Rotation = Mod((n-
1),#out Phi_Num)*10; Peak(n) 'The steps of both Chi tilts & Phi rotations are 10 deg
load_sequential_data(c,Numeric(n+1),n2)
#m_endif
}

macro Sum_up(c, n) {#m_if n<1; 0 #m_else c##_##n + Sum_up(c, Numeric(n-1)) #m_endif}
macro List_of(c, n) {#m_if n==1; c##_##1 #m_else c##_##n, List_of(c, Numeric(n-1)) #m_endif}
prm Average_Peak_Position = (Sum_up(peak_position, #out Total_Num) )/#out Total_Num;:1
prm Average_Peak_Intensity = (Sum_up(peak_intensity, #out Total_Num) )/#out Total_Num;:1
prm Min_Intensity =Min(List_of(peak_intensity,#out Total_Num));:1
prm Max_Intensity =Max(List_of(peak_intensity,#out Total_Num));:1
prm !alpha 1 'Please update the actual incident angle
prm !alpha_c 0.451 'Please update the critical angle for thinfilm (0 to neglect refraction correction)
prm !alpha_r = Sqrt(alpha^2 - alpha_c^2);:0.8925

out "pole_figure.txt" append
Out_String("TEXTURE      4.1\n")
Out_String("FILETYPE     POLEFIGURE\n")
Out_String("VERSION      1\n")
Out_String("TIMESTAMP    2021-04-18T14:36:41.0162969+10:00\n") 'Please update date and time
Out_String("#1   Experimental     ")
out_record out_fmt "%4.1f\t" out_eqn = Average_Peak_Position;
out_record out_fmt "%4.1f" out_eqn = Average_Peak_Position;
out_record out_fmt " °\t"
out_record out_fmt "%4.0f\t" out_eqn = #out Total_Num;
out_record out_fmt "%10.14f\t" out_eqn = Min_Intensity/Average_Peak_Intensity;
out_record out_fmt "%10.14f\t" out_eqn = Max_Intensity/Average_Peak_Intensity;
out_record out_fmt "%10.14f\t" out_eqn = Average_Peak_Intensity;
out_record out_fmt "%7.5f\n" out_eqn = 1.78898; 'Please provide radiation wavelength

'Load 324 peaks using the recursive macro
load_sequential_data(NiFe_Omega_Phi_Prime,1,#out Total_Num)

for xdds {
x_calculation_step 0.1
bkg @ 283.112828 9.446721166
LP_Factor( 0)
Radius(280)
CoKa7_Holzer(0.0001)  }

macro Peak(n) {
```

```
   xo_Is
   xo peak_position##_##n 51.82 'Please provide the initial 2Theta angle
      peak_type spv
         spv_h1 @ 0.4075754746
         spv_h2 @ 0.3601611893
         spv_l1 @ 0.6252098289
         spv_l2 @ 0.6942399704
   I peak_intensity##_##n 128
local Omega = ArcSin(Sin(alpha Deg)/Cos(Chi Deg)) Rad;:1 'Equation(1)
local Phi = ArcSin(Tan(Chi Deg) Tan(alpha Deg) ) Rad;:1 'Equation(2)
local Psi = ArcCos(Cos((peak_position##_##n /2 - Omega) Deg) Cos(Chi Deg)) Rad;:1 'Equation(3)
local Phi1 = ArcCos(Max(-1,Min(1,Sin((peak_position##_##n /2 - Omega)Deg)/Sin(Psi Deg)))) Rad;:1
'Equation(4)
local beta = ArcSin(Sin((peak_position##_##n - Omega) Deg) Cos(Chi Deg)) Rad;:1 'Equation(5)
local Phi2 = ArcTan(Tan((peak_position##_##n - Omega) Deg) Sin(Chi Deg)) Rad;:1 'Equation(6)
local Psi_act = ArcCos((Sin(alpha_r Deg)+Sin(beta Deg))/Sqrt(2 + 2 Sin(alpha_r Deg) Sin(beta Deg) - 2
Cos(alpha_r Deg) Cos(beta Deg) Cos((Phi + Phi2) Deg))) Rad;:1  'Equation(7)
local Phi1_act = ArcCos(Max(-1,Min(1,(Cos(alpha_r Deg)Cos(Phi Deg)-Cos(beta Deg) Cos(Phi2
Deg))/Sqrt((Cos(alpha_r Deg))^2+(Cos(beta Deg))^2-2 Cos(alpha_r Deg) Cos(beta Deg) Cos((Phi + Phi2)
Deg))))) Rad;:1 'Equation(8)
out_record out_fmt "%6.3f\t" out_eqn = Psi_act;
out_record out_fmt "%4.3f\t" out_eqn = 180 - Rotation + Phi1_act - Phi1; '180-Rotation+Equation(8)-
Equation(4)
out_record out_fmt "%10.15f\n" out_eqn =peak_intensity##_##n /Average_Peak_Intensity;}
```

Appendix G

Since version 6, DIFFRAC.TOPAS includes "*num_runs*" and "*Run_Number*" keywords allowing an INP file to be executed multiple times. Combining this function with recursive macros defined in below INP file, the measured d-spacing values in same azimuthal direction are selected and simultaneously fitted, from which a residual stress value together with its refinement error are calculated and reported. The "*Run_Number*" value loops from 0 to "*num_runs*-1", which effectively processes all the azimuthal directions sequentially.

```
'This INP file performs simultaneous refinement sequentially. Residual stress values are refined from
Chi_Num*2 1D peak profiles (.xy) of same hkl for Phi_Num/2 azimuthal directions. D-spacings of Phi and
Phi+180deg are refined together as the - and + sides of same direction. The data files need be named
as "c_n", in which "c" refers to any filename and 'n' refers to a continuous running number. Cf:
X.Wang(2021)J.Appl.Cryst.
num_runs 18
do_errors
'macro Numeric(n) {#prm a = n; #out a} 'v7 TOPAS.inc already included this macro; uncomment this line
when running this INP in v6.

'Load 324 measured peaks (of 9 Chi tilts * 36 Phi directions)
#prm Chi_Num 9 'please provide the number of Chi tilts
#prm Phi_Num 36 'please provide the number of Phi directions
#prm Total_Num =Chi_Num*Phi_Num;

macro load_spaced_data(c,n,n2,sp) '(c=any_filename, n=start_number, n2=finish_number, sp=load data
every sp step)
{#m_if n>n2; #m_else
xdd c##_##n##.xy local Chi = 10 Round((n-(#out Phi_Num +1)/2)/#out Phi_Num); local Rotation = Mod((n-
1),#out Phi_Num)*10; Peak(n) 'The steps of both Chi tilts & Phi rotations are 10 deg
load_spaced_data(c,Numeric(n+sp),n2,sp)
#m_endif
}

macro Sum_up(c, n) {#m_if n<1; 0  #m_else c##_##n + Sum_up(c, Numeric(n-#out Phi_Num/2)) #m_endif}
macro Sum_xy_up(c1,c2,n) {#m_if n<1; 0 #m_else c1##_##n c2##_##n + Sum_xy_up(c1,c2, Numeric(n-#out
Phi_Num/2)) #m_endif}
prm Average_SINPSI2 = (Sum_up(SINPSI2, Numeric(Run_Number+1+#out Phi_Num/2*(#out Chi_Num*2-1))
))/(#out Chi_Num*2);:1
```

```
prm Average_Strain_act = (Sum_up(Strain_act, Numeric(Run_Number+1+#out Phi_Num/2*(#out Chi_Num*2-1))
))/(#out Chi_Num*2);:1
prm Sum_xiyi = Sum_xy_up(Strain_act, SINPSI2, Numeric(Run_Number+1+#out Phi_Num/2*(#out Chi_Num*2-1))
);:1
prm Sum_xixi = Sum_xy_up(SINPSI2, SINPSI2, Numeric(Run_Number+1+#out Phi_Num/2*(#out Chi_Num*2-1))
);:1
prm Slope = (Sum_xiyi - Numeric(#out Chi_Num)*2 * Average_SINPSI2 * Average_Strain_act)/(Sum_xixi -
Numeric(#out Chi_Num)*2 * Average_SINPSI2 * Average_SINPSI2);:1 'Equation(12)
prm !nu 0.32 'Possion ratio
prm !YM 202429 'Young's modulus Unit: MPa
prm rs = Slope / (1+nu) * YM ; :700 'Residual Stress Unit: MPa
prm !alpha 1 'Please update the actual incident angle
prm !alpha_c 0.451 'Please update the critical angle for thinfilm (0 to neglect refraction correction)
prm !alpha_r = Sqrt(alpha^2 - alpha_c^2);:0.8925
prm !d0 2.04613

out "residual_stresses.txt" append
out_record out_fmt "%4.0f\t" out_eqn = 180-Mod(Run_Number,#out Phi_Num)*10;
out_record out_fmt "%10.5f\t" out_fmt_err "%10.5f\n" out_eqn = rs;

'Load 18 peaks along same azimuthal direction using the recursive macro
load_spaced_data(NiFe_Omega_Phi_Prime,Numeric(Run_Number+1),#out Total_Num,Numeric(#out Phi_Num/2))

for xdds {
x_calculation_step 0.05
'exclude 52.9 53.6
bkg @ 283.112828 9.446721166
LP_Factor( 0)
Radius(280)
CoKa7_Holzer(0.0001)  }

macro Peak(n) {
   xo_Is
   xo peak_position##_##n 51.71 'Please provide the initial 2Theta angle
      peak_type spv
         spv_h1 @ 0.4075754746
         spv_h2 @ 0.3601611893
         spv_l1 @ 0.6252098289
         spv_l2 @ 0.6942399704
      I peak_intensity##_##n 128
local Omega = ArcSin(Sin(alpha Deg)/Cos(Chi Deg)) Rad;:1 'Equation(1)
local Phi = ArcSin(Tan(Chi Deg) Tan(alpha Deg) ) Rad;:1 'Equation(2)
local Psi = ArcCos(Cos((peak_position##_##n /2 - Omega) Deg) Cos(Chi Deg)) Rad;:1 'Equation(3)
local Phi1 = ArcCos(Max(-1,Min(1,Sin((peak_position##_##n /2 - Omega)Deg)/Sin(Psi Deg)))) Rad;:1
'Equation (4)
local beta = ArcSin(Sin((peak_position##_##n - Omega) Deg) Cos(Chi Deg)) Rad;:1 'Equation(5)
local Phi2 = ArcTan(Tan((peak_position##_##n - Omega) Deg) Sin(Chi Deg) ) Rad;:1 'Equation(6)
local Psi_act = ArcCos((Sin(alpha_r Deg)+Sin(beta Deg))/Sqrt(2 + 2 Sin(alpha_r Deg) Sin(beta Deg) - 2
Cos(alpha_r Deg) Cos(beta Deg) Cos((Phi + Phi2) Deg))) Rad;:1  'Equation(7)
local Phi1_act = ArcCos(Max(-1,Min(1,(Cos(alpha_r Deg)Cos(Phi Deg)-Cos(beta Deg) Cos(Phi2
Deg))/Sqrt((Cos(alpha_r Deg))^2+(Cos(beta Deg))^2-2 Cos(alpha_r Deg) Cos(beta Deg) Cos((Phi + Phi2)
Deg))))) Rad;:1 'Equation(8)
local Theta2_act = ArcCos(Cos(alpha_r Deg) Cos(beta Deg) Cos((Phi + Phi2) Deg) - Sin(alpha_r Deg)
Sin(beta Deg)) Rad;:1 'Equation(9)
prm SINPSI2##_##n = (Sin(Psi_act Deg))^2;
prm Strain_act##_##n = (Lam/2/Sin(Theta2_act/2 Deg)-d0)/d0;
'Uncomment below lines to export detailed epsilon-SinPsi2 list
'out_record out_fmt "%10.15f\t" out_eqn = (Sin(Psi_act Deg))^2; 'Sin^2Psi
'out_record out_fmt "%10.15f\n" out_eqn = (Lam/2/Sin(Theta2_act/2 Deg)-d0)/d0; 'Actual d-spacing after
refraction correction
}
```